Derivazione errore per il metodo di Simpson

```
> restart:
```

Integrale esatto su intervallo [-h,h]

```
> Iesatto := int( f(x),x=-h..h) ;
```

$$Iesatto := \int_{-h}^{h} f(x)\,dx \qquad (1)$$

Intergrazione approssimata con 1 intervallo (doppio) di ampiezza 2*h col metodo di Simpson

```
> Isimpson := (h/3)*(f(-h)+4*f(0)+f(h)) ;
```

$$Isimpson := \frac{1}{3}\,h\,(f(-h)+4f(0)+f(h)) \qquad (2)$$

```
> Errore := Iesatto - Isimpson ;
```

$$Errore := \int_{-h}^{h} f(x)\,dx - \frac{1}{3}\,h\,(f(-h)+4f(0)+f(h)) \qquad (3)$$

Introduciamo funzione "trucco"

```
> G := int( f(x),x=-t..t) - (t/3)*(f(-t)+4*f(0)+f(t)) - E*(t/h)^N;
```

$$G := \int_{-t}^{t} f(x)\,dx - \frac{1}{3}\,t\,(f(-t)+4f(0)+f(t)) - E\left(\frac{t}{h}\right)^{N} \qquad (4)$$

"E" è l'errore (differenza tra integrale esatto e approssimato) del metodo di Simpson.
Questa espressione va a zero quando t=h, infatti:

```
> subs(t=h,G) ;
```

$$\int_{-h}^{h} f(x)\,dx - \frac{1}{3}\,h\,(f(-h)+4f(0)+f(h)) - E \qquad (5)$$

"E" è l'errore, cioè quel numero che manda a zero la precedente espressione.
Quando t=0 abbiamo:

```
> eval(subs(t=0,G));
```

$$0 \qquad (6)$$

Quindi la funzione G si azzera in 0 ed h, esiste punto 0<alpha<h tale che G'(alpha) = 0

```
> DG := simplify(diff(G,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DG := \frac{2}{3}f(t) + \frac{2}{3}f(-t) - \frac{4}{3}f(0) + \frac{1}{3}\,t\,\mathrm{D}(f)(-t) - \frac{1}{3}\,t\,\mathrm{D}(f)(t) - E\,t^{N-1}\,h^{-N}\,N \qquad (7)$$

Valutando la funzione DG in 0

```
> subs(t=0,DG) ;
```

$$0 \qquad (8)$$

Quindi la funzione G' si azzera in 0 ed alpha, esiste punto 0<beta<alpha<h tale che G"(beta) = 0.

```
> DDG := simplify(diff(DG,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DDG := \frac{1}{3}\mathrm{D}(f)(t) - \frac{1}{3}\mathrm{D}(f)(-t) - \frac{1}{3}\,t\,\mathrm{D}^{(2)}(f)(-t) - \frac{1}{3}\,t\,\mathrm{D}^{(2)}(f)(t) - E\,t^{N-2}\,h^{-N}\,N^{2} + E\,t^{N-2}\,h^{-N}\,N \qquad (9)$$

```
> solve(DDG,E) ;
```

$$\frac{1}{3}\,\frac{\mathrm{D}(f)(t)-\mathrm{D}(f)(-t)-t\,\mathrm{D}^{(2)}(f)(-t)-t\,\mathrm{D}^{(2)}(f)(t)}{N\,t^{N-2}\,h^{-N}\,(N-1)} \tag{10}$$

Quando t=0 abbiamo:

```
> eval(subs(t=0,DDG));
```

$$0 \tag{11}$$

Quindi la funzione G" si azzera in 0 ed beta, esiste punto 0<omega<beta<alpha<h tale che G'''(omega) = 0.

```
> DDDG := simplify(diff(DDG,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DDDG := \frac{1}{3}\,t\,\mathrm{D}^{(3)}(f)(-t)-\frac{1}{3}\,t\,\mathrm{D}^{(3)}(f)(t)-E\,t^{N-3}\,h^{-N}\,N^3+3\,E\,t^{N-3}\,h^{-N}\,N^2-2\,E\,t^{N-3}\,h^{-N}\,N \tag{12}$$

La differenza

$$\frac{\left(\mathrm{D}^{(3)}(f)(t)-\mathrm{D}^{(3)}(f)(-t)\right)}{(2\,t)}=\mathrm{D}^{(4)}(f)(z)\ \left(\textit{dal teorema di Lagrange}\ f(b)-f(a)=f'(c)(b-a)\right)$$

```
> EQ1 := simplify(subs((D@@3)(f)(t)=(D@@4)(f)(z(t))*2*t+(D@@3)(f)(-t),DDDG)) ;
```

$$EQ1 := -\frac{2}{3}\,\mathrm{D}^{(4)}(f)(z(t))\,t^2-E\,t^{N-3}\,h^{-N}\,N^3+3\,E\,t^{N-3}\,h^{-N}\,N^2-2\,E\,t^{N-3}\,h^{-N}\,N \tag{13}$$

Ponendo N=5

```
> EQ2 := subs(N=5,EQ1) ;
```

$$EQ2 := -\frac{2}{3}\,\mathrm{D}^{(4)}(f)(z(t))\,t^2-\frac{60\,E\,t^2}{h^5} \tag{14}$$

Questa espressione è zero quando t=omega, chiamo eta = z(omega)

```
> EQ3 := subs(t=omega,z(omega)=eta,EQ2);
```

$$EQ3 := -\frac{2}{3}\,\mathrm{D}^{(4)}(f)(\eta)\,\omega^2-\frac{60\,E\,\omega^2}{h^5} \tag{15}$$

Poiché omega != 0 possiamo risolvere per l'errore E

```
> solve( EQ3, {E} ) ;
```

$$\left\{E=-\frac{1}{90}\,\mathrm{D}^{(4)}(f)(\eta)\,h^5\right\} \tag{16}$$

Questo è l'errore locale per un singolo intervallone di ampiezza H (H=2*h, o coppia di intervalli)